

Determining $n(T)$ for Extrinsic Materials
by Dr. Colton, Physics 581 (last updated: Fall 2021)

Intrinsic reminder

As a reminder, for intrinsic materials we were able to deduce the number of electrons in the conduction band as a function of temperature, $n(T)$ for a given material (with a given band gap), by writing these equations for n and p , and using the equation $n = p$ to connect the two.

$$n = 2 \left(\frac{m_e kT}{2\pi\hbar^2} \right)^{3/2} e^{-(E_c - \mu)/kT} \quad (1)$$

$$p = 2 \left(\frac{m_h kT}{2\pi\hbar^2} \right)^{3/2} e^{-(\mu - E_v)/kT} \quad (2)$$

$$n = p \quad (\text{intrinsic condition}) \quad (3)$$

Those were the three governing equations for the intrinsic case. E_c and E_v are the conduction and valence band energies, respectively.

None of the equations individually gives enough information to solve for $n(T)$, because we don't know how μ varies with temperature (at least, we didn't at first). However, multiplying (1) and (2) and taking the square root allowed us to determine $n(T)$ by recognizing that $n^2 = np$ from the intrinsic condition. We were also able to determine $\mu(T)$ by plugging (1) and (2) into (3) and solving for μ .

Alternatively, we could have first solved for $\mu(T)$ and then plugged it back into (1). That would have led to the same result for $n(T)$ that we obtained through the other method. We'll take that second approach for extrinsic (doped) materials: to find $n(T)$ we will first find μ , then plug μ into (1) to get n .

Governing equations for extrinsic materials

Let's now consider an n-type doped material; hopefully the extrapolation to a p-type material will be clear should you ever need to do that. For an n-type material, (1) and (2) for n and p are still the same,¹ but the equation connecting the two is now different:

$$n = p + N_d^+ \quad (\text{extrinsic condition}) \quad (4)$$

Here N_d^+ is the concentration of ionized donors, i.e. the number of donors per volume whose electrons have been promoted to the conduction band. A formula for N_d^+ can be obtained via a thermodynamic treatment of ionized donors, as the following:²

¹ This is actually an approximation, since technically in deriving (1) and (2) we had to make a small temperature approximation which may or may not exactly apply now. But we'll ignore that, since everyone else does, and the resulting equations work well.

² See for example, problem 7.5 in Daniel Schroeder, *An Introduction to Thermal Physics*; and also these websites <https://lampx.tugraz.at/~hadley/psd/problems/ionizeddonors/Q.php> and <http://ecee.colorado.edu/~bart/book/extrinsi.htm> (under "2.7.4.2 General analysis").

$$N_d^+ = \frac{N_d}{1 + 2e^{(\mu - E_d)/kT}} \quad (5)$$

where N_d is the total concentration of donors and E_d is the absolute energy of the donor level (i.e. not the donor ionization energy, which is the difference between E_c and E_d).

Equations (1), (2), (4), and (5) are the governing equations for the extrinsic case.

Solving the governing equations for the extrinsic case

We can now put the four governing equations together by plugging (1), (2), and (5) into (4), like this:

$$2 \left(\frac{m_e kT}{2\pi\hbar^2} \right)^{3/2} e^{-(E_c - \mu)/kT} = 2 \left(\frac{m_h kT}{2\pi\hbar^2} \right)^{3/2} e^{-(\mu - E_v)/kT} + \frac{N_d}{1 + 2e^{(\mu - E_d)/kT}} \quad (6)$$

For a situation where we are given the material and the amount of doping, μ and T are the only unknowns in that equation. The equation can therefore in theory then be solved for μ as a function of T , and $\mu(T)$ can then be plugged into (1) to give us $n(T)$.

However, unlike the intrinsic case, (6) is not analytically solvable for $\mu(T)$. Therefore we must solve it numerically: pick a temperature, solve numerically for μ at that temperature, then plug that μ into (1) to get n for that temperature. Let's do that in an example.

Example 1: Silicon Doped With 10^{21} Donors/m³, T = 77 and 300 K

For silicon at 300 K, the appropriate “density of states effective masses” for electrons and holes are $1.091m$ and $0.999m$, and the bandgap is 1.1242 eV.³ Therefore the quantity $2 \left(\frac{m_e kT}{2\pi\hbar^2} \right)^{3/2}$ equals $2.85962 \times 10^{25} \text{ m}^{-3}$ and the quantity $2 \left(\frac{m_h kT}{2\pi\hbar^2} \right)^{3/2}$ equals $2.50565 \times 10^{25} \text{ m}^{-3}$. To simplify typing things into Mathematica, I'm therefore going to rewrite (1) and (2) like this:

$$n = 2.85962 \times 10^{25} \left(\frac{T}{300} \right)^{3/2} e^{-(E_c - \mu)/kT}$$

$$p = 2.50565 \times 10^{25} \left(\frac{T}{300} \right)^{3/2} e^{-(\mu - E_v)/kT}$$

where the temperatures must now be explicitly given in kelvins and the concentrations n and p are in m^{-3} .

I'm also going to set the valence band energy to zero, as is often done, so $E_v = 0$, $E_c = \text{band gap}$, and $E_d = E_g - 0.045$ eV (taking the donor ionization energy for silicon to be 45 meV). Remember, the procedure is to use (6) to solve for μ for the given T, then plug that μ back into (1) to get n . I'll now turn to Mathematica to first solve for 77 K and then for 300 K. Pay attention to the comments in the code.

³ These values will shift with temperature. However, for simplicity in the examples that follow I'll continue to use these 300 K values even though the masses will go up and energy gap will go down with increasing temperature because the lattice undergoes thermal expansion which affects the band structure.

Setting things up

```

In[10]:= k = 1.38064852*^-23;
         e = 1.60217662*^-19;
         Eg = 1.1242 e; (* band gap of Si in eV, multiplying by e to convert to J *)
         Ev = 0; (* picking my zero of energy to be at VB *)
         Ec = Ev + Eg;
         Ed = Ec - 0.045 e; (* the absolute position of donor level, not relative *)
         Nd = 1*^21;

(* I'm going to multiply mu by charge of electron in these next equations so mu in the plot
will have units of eV. I will also divide all concentrations by 10^21 for now so the
"FindRoot" command doesn't have to work with huge exponents. *)
n[mu_, T_] := 2.85962*^25 (T / 300) ^ (3 / 2) Exp[(mu e - Ec) / (k T)] / 1*^21
p[mu_, T_] := 2.50565*^25 (T / 300) ^ (3 / 2) Exp[(Ev - mu e) / (k T)] / 1*^21
Ndplus[mu_, T_] := Nd / (1 + 2 Exp[(mu e - Ed) / (k T)]) / 1*^21 (* mu in eV, n in 10^21 *)

(* The extrinsic condition is "n = p + Ndplus". I'm going to plot both sides of that equation
as a function of mu then find the mu where they intersect. Then I'll use that mu in the n
equation to get the answer for a desired temperature. *)


```

Solving when T = 77K

```

In[17]:= T = 77;
Plot[{n[mu, T], p[mu, T] + Ndplus[mu, T]}, {mu, 0, 1.2}]
(* plotting both sides of the extrinsic equation *)

```



```

In[19]:= (* Looks like answer must be around mu = 1.05 *)
         FindRoot[n[mu, T] == p[mu, T] + Ndplus[mu, T], {mu, 1.05}]

```

```
Out[19]:= {mu -> 1.06765}
```

First step: $\mu(T = 77\text{K}) = 1.06765 \text{ eV}$

```

In[20]:= foundmu = mu /. %; (* This is needed because the FindRoot command provides its answer in a
weird format *)
         n[foundmu, T] * 1*^21 (* the answer for T=77! *)


```

```
Out[21]:= 7.40182 * 10^20
```

Second step: $n(T = 77\text{K}) = 7.40182 \times 10^{20} \text{ m}^{-3}$

Solving when T = 300K

```
In[22]:= (* I'll do everything in one cell now. I'll try to use the same starting point for my in
the FindRoot command that I did for 77 K, namely 1.05. It could fail... but fortunately it doesn't. *)
T = 300;
Plot[{n[mu, T], p[mu, T] + Ndplus[mu, T]}, {mu, 0, 1.2}]
FindRoot[n[mu, T] == p[mu, T] + Ndplus[mu, T], {mu, 1.05}]
foundmu = mu /. %;
n[foundmu, T] * 1*^21 (* the answer for T=300! *)
```



Out[24]= {mu → 0.858922}

First step: $\mu(T = 300\text{K}) = 0.858922 \text{ eV}$

Out[26]= 9.99602×10^{20}


Second step: $n(T = 300\text{K}) = 9.99602 \times 10^{20} \text{ m}^{-3}$

Example 2: Silicon Doped With 10^{21} Donors/ m^3 , all temperatures

That same procedure can be employed to determine μ for any arbitrary temperature. I'll do all temperatures at the same time now, letting Mathematica do the "pick a temperature" part automatically.


Solving for arbitrary temperatures

```
In[27]:= Off[General::munfl] (* This suppresses a warning message that will otherwise show up *)
(* I'm going to skip making the plot of the two sides of the extrinsic equation for the
desired temperature and will skip straight to the FindRoot command *)
actualmu[T_] := mu /. FindRoot[n[mu, T] == p[mu, T] + Ndplus[mu, T], {mu, 1.05}];
Plot[actualmu[T], {T, 5, 1000}, PlotRange → {0, 1.2}]
```



Plot of μ in eV as a function of temperature in K, for Si with $N_d = 10^{21} \text{ m}^{-3}$

```
In[30]:= actualn[T_] := n[actualmu[T], T] * 10^21
LogPlot[actualn[T], {T, 15, 1000}, PlotRange -> {{0, 1000}, {10^19, 10^23}}]
```



Plot of n in m^{-3} as a function of temperature in K, for Si with $N_d = 10^{21} \text{ m}^{-3}$

This is the main result of this handout!

Approximate results for the three temperature regimes

As you can tell from the last plot, there are three distinct types of behavior for low, medium, and high temperatures. They have to do with how the terms in (4) compare in size. It's fairly easy to get approximate results for the three regimes.

I'll now use the notation that the non-exponential parts of (1) and (2) can be written as n_0 and p_0 :

$$n_0 = 2 \left(\frac{m_e kT}{2\pi \hbar^2} \right)^{3/2}$$

$$p_0 = 2 \left(\frac{m_h kT}{2\pi \hbar^2} \right)^{3/2}$$

Low temperature regime: $n \approx N_d^+$

At really low temperatures, essentially none of the electrons in the valence band are thermally excited into the conduction band: $p \approx 0$. Therefore all of the conduction electrons come from the ionized donors and (4) becomes:

$$n = \frac{N_d}{1 + 2e^{(\mu - E_d)/kT}} \quad (7)$$

If the temperature is low enough, then $(\mu - E_d)/kT$ is large, and the exponential in the denominator of (7) is much larger than 1.

$$n = \frac{N_d}{2e^{(\mu - E_d)/kT}}$$

$$n = \frac{N_d}{2} e^{-(\mu - E_d)/kT}$$

We can multiply the left and right hand sides by the left and right hand sides of (1) to remove μ from the equation, then take the square root of n^2 to obtain n :

$$\begin{aligned}
n^2 &= \frac{N_d}{2} e^{-(\mu - E_d)/kT} \times n_0 e^{-(E_c - \mu)/kT} \\
n^2 &= \frac{n_0 N_d}{2} e^{-(E_c - E_d)/kT} \\
n &= \sqrt{\frac{n_0 N_d}{2}} e^{-E_{d, \text{binding}}/2kT} \tag{8}
\end{aligned}$$

This is Kittel's Eq. 8.53 (but he doesn't derive it), although he is missing a factor of $\sqrt{2}$ in his equation. This is called the "freeze out" regime because typically n will be quite small.

Medium temperature regime (includes 300 K): $n \approx N_d$

As the temperature increases, all of the donors get ionized while the valence band electrons are still not able to reach the conduction band. We still have $p \approx 0$, but the value of N_d^+ is now extremely close to N_d . Therefore for this regime we simply have

$$n = N_d \tag{9}$$

Another way of looking at this regime is that (7) still applies, but the exponential in the denominator of (7) is now much less than one because $(\mu - E_d)/kT$ has switched from positive to negative.

High temperature regime: $n \approx p$

At high enough temperatures, thermal excitation of the electrons from the valence band to the conduction band becomes significant. Therefore $p \gg N_d^+$ in (5) and we have:

$$n \approx p$$

That is exactly the condition we already solved for intrinsic semiconductors, and indeed this is called the "intrinsic regime". The result we previously obtained was:

$$n = \sqrt{n_0 p_0} e^{-E_g/2kT} \tag{10}$$

Equations (8), (9), and (10) can be used to approximate the true results for the three temperature regimes.